Universidade Estadual de Maringá
Departamento de Matemática
Prof. Doherty Andrade (DMA- UEM)
TOPICOS EM EDO - CURSO DE VERAO -DMA/UEM

O Comando dsolve

O comando básico do Maple para resolver equações diferenciais ordinarias eh o " dsolve ".

A sintaxe do dsolve é

dsolve("oque", "como");

O "oque" refere-se a EDO (ou sistema de edo's) junto com condições iniciais. O "como" especifica qual rotina do Maple vai ser utilizada.

É útil dar nomes a todas as equações e condições iniciais quando usamos o dsolve . Exemplos:

```
> eq:=diff(y(x),x)=x*y(x);
```

$$eq := \frac{\partial}{\partial x} y(x) = x\, y(x)$$

```
> init:=y(2)=1;
```

$$init := y(2) = 1$$

Agora, eq é o nome da edo que queremos resolver e init é o nome das condições iniciais. É

importante usar y(x) -- isto indica que tomamos y como uma variavel dependente e x como variavel independente.
Para resolver sem a condição inicial, isto é achar a solução geral, usamos `dsolve`

```
> dsolve(eq,y(x));
```

$$y(x) = e^{\left(\frac{1}{2} x^2\right)} \_C1$$

Note que _C1 é a constante arbitraria produzida pelo maple.

Para resolver o PVI devemos agrupar a EDO junto com a condicao inicial entre chaves.

```
> dsolve({eq,init},y(x));
```

$$y(x) = \frac{e^{\left(\frac{1}{2}x^2\right)}}{e^2}$$

As vezes Maple dá uma resposta implicita, assim é comum pedir que o Maple explicite a resposta em função da variavel dependente. Exemplo:

```
> dsolve(diff(y(x),x)=y(x)^2, y(x));
```

$$\frac{1}{y(x)} = -x + \_C1$$

Usamos o comando explicit como opcao em dsolve :

```
> dsolve(diff(y(x),x)=y(x)^2, y(x), explicit);
```

$$y(x) = -\frac{1}{x - \_C1}$$

Assim é melhor!!!

Uso mais avançado de dsolve:

Existem 4 maneiras de usar o dsolve para situações além das edos de primeira ordem e pvi simples.

1. **Equações de ordem superior**

2. **Sistemas de EDO's**

3. **Soluções numericas**

4. **Usando séries de potências para obter soluções de edo's**

1. **Equações de ordem superior** : edo's de ordem 2 ou superior podem ser resolvidas usando o dsolve . As derivadas de ordem superior são escritas como no exemplo.

```
> diff(f(x),x$3);
```

$$\frac{\partial}{\partial x}\left(\frac{\partial}{\partial x}\left(\frac{\partial}{\partial x}f(x)\right)\right)$$

Uma EDO de ordem 2:

> **eqn2:=diff(y(x),x$2)+3*diff(y(x),x)+2*y(x)=exp(x);**

$$eqn2 := \left(\frac{\partial}{\partial x}\left(\frac{\partial}{\partial x}\mathrm{y}(x)\right)\right) + 3\left(\frac{\partial}{\partial x}\mathrm{y}(x)\right) + 2\,\mathrm{y}(x) = \mathbf{e}^{x}$$

A solução geral é:

> **dsolve(eqn2,y(x));**

$$\mathrm{y}(x) = \frac{1}{6}\mathbf{e}^{x} + \_C1\,\mathbf{e}^{(-2\,x)} + \_C2\,\mathbf{e}^{(-x)}$$

Para este problema vamos especificar condições de y em x=1:

> **inits:=y(1)=2, D(y)(1)=4;**

$$inits := \mathrm{y}(1) = 2, \mathrm{D}(y)(1) = 4$$

> **dsolve({eqn2,inits},y(x));**

$$\mathrm{y}(x) = \frac{1}{6}\,\frac{(\mathbf{e}^{x})^{2} + 6\left(\frac{1}{3}(\mathbf{e})^{3} - 6\,(\mathbf{e})^{2}\right)\mathbf{e}^{(-2\,x)}\,\mathbf{e}^{x} - 3\,(\mathbf{e})^{2} + 48\,\mathbf{e}}{\mathbf{e}^{x}}$$

2. **Sistemas de EDO's** : Vamos ver um exemplo.

> **eqns:=diff(y(x),x)+diff(z(x),x)=x, diff(y(x),x)-2*diff(z(x),x)=x^2;**

$$eqns := \left(\frac{\partial}{\partial x}\mathrm{y}(x)\right) + \left(\frac{\partial}{\partial x}\mathrm{z}(x)\right) = x, \left(\frac{\partial}{\partial x}\mathrm{y}(x)\right) - 2\left(\frac{\partial}{\partial x}\mathrm{z}(x)\right) = x^{2}$$

> **inits:= y(0)=1, z(0)=2;**

$$inits := \mathrm{y}(0) = 1, \mathrm{z}(0) = 2$$

> **dsolve({eqns,inits},{y(x),z(x)});**

$$\{\mathrm{y}(x) = 1 + \frac{1}{3}x^{2} + \frac{1}{9}x^{3}, \mathrm{z}(x) = 2 + \frac{1}{6}x^{2} - \frac{1}{9}x^{3}\}$$

3. **Soluções Numéricas** : Em geral é impossivel obter explicitamente a soluçao de uma EDO, neste caso usamos um método numérico para aproximar a solução. O Maple faz isto, basta usar a opcao numeric em dsolve , como no exemplo:

> **eqn:=diff(y(x),x)+exp(y(x))*x^3=2*sin(x); init:=y(0)=2;**

$$eqn := \left(\frac{\partial}{\partial x}y(x)\right) + e^{y(x)}\,x^3 = 2\sin(x)$$

$$init := y(0) = 2$$

> **F:=dsolve({eqn,init},y(x),numeric);**

$$F := \mathbf{proc}(rkf45\_x)\ \ldots\ \mathbf{end}$$

O valor de F em x=2 é

> **F(2);**

$$[x = 2, y(x) = -.7815970926289340]$$

Em geral é util plotar numericamente a solução obtida de uma EDO. Para fazer isto usamos o comando odeplot **ao resultado (F neste caso) de** dsolve(....,numeric) **. Para usar**

odeplot precisamos de "chamar" o plots **.**

> **with(plots,odeplot);**

$$[odeplot]$$

Para plotar a solucao usamos a sintaxe.

> **odeplot(F,[x,y(x)],-2..2);**



4 . Usando series de potencias. Exemplo: resolver y'+x*y=0, y(0)=1 por series .

> **dsolve({diff(y(x),x)+x*y(x)=0, y(0)=1},y(x), series);**

$$y(x) = 1 - \frac{1}{2} x^2 + \frac{1}{8} x^4 + O(x^6)$$

Se voce quer mais termos, peça mais.

> **Order:=14: dsolve({diff(y(x),x)+x*y(x)=0, y(0)=1},y(x), series);**

$$y(x) = 1 - \frac{1}{2} x^2 + \frac{1}{8} x^4 - \frac{1}{48} x^6 + \frac{1}{384} x^8 - \frac{1}{3840} x^{10} + \frac{1}{46080} x^{12} + O(x^{14})$$

>